Harmonic expansion of the metric tensor

Preliminaries

Load tensor package

In[1]:=
```
<< xAct`xPert`
```

```
------------------------------------------------------------
Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}
CopyRight (C) 2003-2020, Jose M. Martin-Garcia, under the General Public License.
Connecting to external linux executable...
Connection established.
------------------------------------------------------------
Package xAct`xTensor`  version 1.1.4, {2020, 2, 16}
CopyRight (C) 2002-2020, Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
Package xAct`xPert`  version 1.0.6, {2018, 2, 28}
CopyRight (C) 2005-2018, David Brizuela, Jose M. Martin-Garcia
  and Guillermo A. Mena Marugan, under the General Public License.
------------------------------------------------------------
These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.
------------------------------------------------------------
** Variable $PrePrint assigned value ScreenDollarIndices
** Variable $CovDFormat changed from Prefix to Postfix
** Option AllowUpperDerivatives of ContractMetric changed from False to True
** Option MetricOn of MakeRule changed from None to All
** Option ContractMetrics of MakeRule changed from False to True
```

Nicer printing

In[2]:=
```
$PrePrint = ScreenDollarIndices;
```

In[3]:=
```
$CovDFormat = "Prefix";
```

Object definitions

Spacetime manifold

The spacetime manifold M, on which tensors will be defined. Some Greek letters are defined as tangent space indices.

In[4]:=
```
DefManifold[M, 4, {α, β, γ, μ, ν, ρ, σ, ω, τ, θ, ϕ, κ, ξ, ζ, ϑ, φ}]
```

```
** DefManifold : Defining manifold M.
** DefVBundle : Defining vbundle TangentM .
```

Metric

The metric g of signature (-,+,+,+). The Levi-Civita derivative of a tensor A_μ will be written as $\nabla_\mu A_\nu$ in prefix notation or $A_{\nu;\mu}$ in postfix notation.

In[5]:=
```
DefMetric[{3, 1, 0}, Met[-μ, -ν], CD, {";", "∇"}, PrintAs → "g"]
```

```
** DefTensor : Defining symmetric metric tensor Met[-μ, -ν].
** DefTensor : Defining antisymmetric tensor epsilonMet [-α, -β, -γ, -ζ].
** DefTensor : Defining tetrametric TetraMet [-α, -β, -γ, -ζ].
** DefTensor : Defining tetrametric TetraMet† [-α, -β, -γ, -ζ].
** DefCovD : Defining covariant derivative CD[-μ].
** DefTensor : Defining vanishing torsion tensor TorsionCD [α, -β, -γ].
** DefTensor : Defining symmetric Christoffel tensor ChristoffelCD [α, -β, -γ].
** DefTensor : Defining Riemann tensor RiemannCD [-α, -β, -γ, -ζ].
** DefTensor : Defining symmetric Ricci tensor RicciCD [-α, -β].
** DefCovD : Contractions of Riemann automatically replaced by Ricci .
** DefTensor : Defining Ricci scalar RicciScalarCD [].
** DefCovD : Contractions of Ricci automatically replaced by RicciScalar .
** DefTensor : Defining symmetric Einstein tensor EinsteinCD [-α, -β].
** DefTensor : Defining Weyl tensor WeylCD [-α, -β, -γ, -ζ].
** DefTensor : Defining symmetric TFRicci tensor TFRicciCD [-α, -β].
** DefTensor : Defining Kretschmann scalar KretschmannCD [].
** DefCovD : Computing RiemannToWeylRules for dim 4
** DefCovD : Computing RicciToTFRicci for dim 4
** DefCovD : Computing RicciToEinsteinRules for dim 4
** DefTensor : Defining weight +2 density DetMet []. Determinant .
```

We will work with tensor densities. This will make sure that the covariant derivative acts correctly on densities.

In[6]:=
```
WeightedWithBasis[CD] ^= AIndex;
```

Perturbations of the metric are written as δg.

In[7]:=
```
DefMetricPerturbation[Met, δMet, ϵ, PrintAs → "δg"]
```

```
** DefParameter : Defining parameter ϵ.
** DefTensor : Defining tensor δMet[LI[order], -α, -β].
```

Inverse metric density

This defines the “gothic metric”, or densitized inverse metric $\mathfrak{g}^{\mu\nu} = g^{\mu\nu}\sqrt{-g}$.

In[8]:=
```
DefTensor[DIMet[μ, ν], M, Symmetric[{1, 2}], WeightOfTensor → AIndex, PrintAs → "𝔤"]
```

```
** DefTensor : Defining tensor DIMet[μ, ν].
```

We then define the metric perturbation as $h^{\mu\nu} = \mathfrak{g}^{\mu\nu} - \eta^{\mu\nu}$.

In[9]:=
```
DefTensorPerturbation[hh[LI[0], μ, ν], DIMet[μ, ν], {M}, Symmetric[{2, 3}], PrintAs → "h"]
```

```
** DefTensor : Defining tensor hh[LI[order], μ, ν].
```

For the inverse of the gothic metric, we write $\bar{\mathfrak{g}}_{\mu\nu}$.

In[10]:=
```
DefTensor[DMet[-μ, -ν], M, Symmetric[{1, 2}], WeightOfTensor → -AIndex, PrintAs → "𝔤̄"]
```

```
** DefTensor : Defining tensor DMet[-μ, -ν].
```

Finally, it will be convenient to define $\bar{h}_{\mu\nu} = \bar{\mathfrak{g}}_{\mu\nu} - \eta_{\mu\nu}$.

In[11]:=
```
DefTensorPerturbation[ihh[LI[0], -μ, -ν],
 DMet[-μ, -ν], {M}, Symmetric[{2, 3}], PrintAs → "h̄"]
```

```
** DefTensor : Defining tensor ihh[LI[order], -μ, -ν].
```

Minkowski metric

We also define the Minkowski metric as a flat metric:

In[12]:=
```
DefMetric[{3, 1, 0}, Mink[-μ, -ν], PD, {",", "∂"}, FlatMetric → True, PrintAs → "η"]
```

```
••• DefMetric : There are already metrics {Met} in vbundle TM.
** DefTensor : Defining symmetric metric tensor Mink[-μ, -ν].
** DefTensor : Defining inverse metric tensor InvMink[μ, ν]. Metric is frozen!
** DefTensor : Defining antisymmetric tensor epsilonMink[-α, -β, -γ, -ζ].
** DefTensor : Defining tetrametric TetraMink[-α, -β, -γ, -ζ].
** DefTensor : Defining tetrametric TetraMink† [-α, -β, -γ, -ζ].
** DefMetric : Associating fiducial flat derivative PD to metric.
** DefTensor : Defining weight +2 density DetMink[]. Determinant.
```

Energy-momentum tensor

For working with the Einstein equations, we need the energy-momentum tensor.

In[13]:=
```
DefTensor[EnMom[μ, ν], M, Symmetric[{1, 2}], PrintAs → "T"]
```
```
** DefTensor : Defining tensor EnMom[μ, ν].
```

Gravitational constant

This is the gravitational constant, which also enters the Einstein equations.

In[14]:=
```
DefConstantSymbol[NewtG, PrintAs → "G"]
```
```
** DefConstantSymbol : Defining constant symbol NewtG.
```

Tensors in densitized language

Metric

We start be defining a few helpful rules, which will be used later through the calculations. First, we declare that $\bar{\mathfrak{g}}_{\mu\nu}$ shall be the inverse of $\mathfrak{g}^{\mu\nu}$. Hence, contractions will evaluate to the Kronecker symbol.

In[15]:=
```
ggru = Union[Flatten[MakeRule[#, MetricOn → None, ContractMetrics → False] &/@
    {{DIMet[μ, ρ] × DMet[-ν, -ρ], delta[-ν, μ]}, {DIMet[ρ, μ] × DMet[-ρ, -ν], delta[-ν, μ]},
     {DIMet[ρ, μ] × DMet[-ν, -ρ], delta[-ν, μ]}, {DIMet[μ, ρ] × DMet[-ρ, -ν], delta[-ν, μ]}}]]
```

Out[15]=
$$\left\{\text{HoldPattern}\left[\tilde{\mathfrak{g}}^{\mu\rho}\,\bar{\mathfrak{g}}_{\nu\rho}\right] :\to \text{Module}\left[\{\}, \delta_{\nu}{}^{\mu}\right], \text{HoldPattern}\left[\tilde{\mathfrak{g}}^{\rho\mu}\,\bar{\mathfrak{g}}_{\nu\rho}\right] :\to \text{Module}\left[\{\}, \delta_{\nu}{}^{\mu}\right], \text{HoldPattern}\left[\tilde{\mathfrak{g}}^{\mu\rho}\,\bar{\mathfrak{g}}_{\rho\nu}\right] :\to \text{Module}\left[\{\}, \delta_{\nu}{}^{\mu}\right], \text{HoldPattern}\left[\tilde{\mathfrak{g}}^{\rho\mu}\,\bar{\mathfrak{g}}_{\rho\nu}\right] :\to \text{Module}\left[\{\}, \delta_{\nu}{}^{\mu}\right]\right\}$$

Another consequence is that the derivative of the lower index gothic metric is expressed in terms of derivatives of the upper index version. This is given by the following rule.

In[16]:=
```
dgru = MakeRule[{PD[-ρ][DMet[-μ, -ν]], -DMet[-μ, -α] × DMet[-ν, -β] × PD[-ρ][DIMet[α, β]]},
  MetricOn → None, ContractMetrics → False]
```

Out[16]=
$$\left\{\text{HoldPattern}\left[\partial_{\rho}\bar{\mathfrak{g}}_{\mu\nu}\right] :\to \text{Module}\left[\{\alpha, \beta\}, -\bar{\mathfrak{g}}_{\mu\alpha}\,\bar{\mathfrak{g}}_{\nu\beta}\,\partial_{\rho}\tilde{\mathfrak{g}}^{\alpha\beta}\right]\right\}$$

Christoffel symbols

Now we will calculate the Christoffel symbols in terms of the gothic metric. This can be done using the fact that the gothic metric is covariantly constant, just like the normal metric. Hence, the following expression vanishes, also after lowering indices. Also note the appearance of the trace of the Christoffel symbol, since the gothic metric is a tensor density and not a tensor.

In[17]:=
```
CD[-ρ][DIMet[α, β]]
DMet[-μ, -α] × DMet[-ν, -β] %;
ChangeCovD[%, CD, PD];
Expand[%];
del1 = % //. ggru
```

Out[17]= $\nabla_\rho \tilde{g}^{\alpha\beta}$

Out[21]= $\Gamma[\nabla]^{\alpha}{}_{\rho\nu}\ \bar{g}_{\mu\alpha} - \Gamma[\nabla]^{\alpha}{}_{\rho\alpha}\ \bar{g}_{\mu\nu} + \Gamma[\nabla]^{\beta}{}_{\rho\mu}\ \bar{g}_{\nu\beta} + \bar{g}_{\mu\alpha}\ \bar{g}_{\nu\beta}\ \partial_\rho \tilde{g}^{\alpha\beta}$

To derive a formula for a single Christoffel symbol, we use the same trick as when calculating them in terms of the metric. The following recombination of indices gives us just one Christoffel symbol, plus some traces.

In[22]:=
```
del1 - 2 Antisymmetrize[del1, {-μ, -ρ}] - 2 Antisymmetrize[del1, {-ν, -ρ}];
del2 = ToCanonical[%, UseMetricOnVBundle → None]
```

Out[23]= $\Gamma[\nabla]^{\alpha}{}_{\rho\alpha}\ \bar{g}_{\mu\nu} - \Gamma[\nabla]^{\alpha}{}_{\nu\alpha}\ \bar{g}_{\mu\rho} - \Gamma[\nabla]^{\alpha}{}_{\mu\alpha}\ \bar{g}_{\nu\rho} + 2\ \Gamma[\nabla]^{\alpha}{}_{\mu\nu}\ \bar{g}_{\rho\alpha} + \bar{g}_{\nu\alpha}\ \bar{g}_{\rho\beta}\ \partial_\mu \tilde{g}^{\alpha\beta} + \bar{g}_{\mu\alpha}\ \bar{g}_{\rho\beta}\ \partial_\nu \tilde{g}^{\alpha\beta} - \bar{g}_{\mu\alpha}\ \bar{g}_{\nu\beta}\ \partial_\rho \tilde{g}^{\alpha\beta}$

Next, we must eliminate the traces. By contracting two indices from the (vanishing) expression above, we get that also the following vanishes.

In[24]:=
```
-DIMet[ν, ρ] del2 / 2;
Expand[%];
% //. ggru;
del3 = ToCanonical[%, UseMetricOnVBundle → None]
```

Out[27]= $\Gamma[\nabla]^{\alpha}{}_{\mu\alpha} - \frac{1}{2}\ \bar{g}_{\alpha\beta}\ \partial_\mu \tilde{g}^{\alpha\beta}$

Now this is the desired relation that gives us the trace. We use it to eliminate the trace from the last equation, and finally get a solution for the Christoffel symbol.

In[28]:=
```
DMet[-ν, -ρ] del3 + DMet[-μ, -ρ] (delta[-ν, μ] del3) -
   DMet[-μ, -ν] (delta[-ρ, μ] del3) + del2;
DIMet[ρ, σ] % / 2;
Expand[%];
% //. ggru;
chru = MakeRule[{ChristoffelCD[σ, -μ, -ν], Evaluate[ChristoffelCD[σ, -μ, -ν] - %]},
   MetricOn → None, ContractMetrics → False];
(# == (# /. chru)) &[ChristoffelCD[ρ, -μ, -ν]]
```

Out[32]= $\Gamma[\nabla]^{\rho}{}_{\mu\nu} == -\frac{1}{4}\ \tilde{g}^{\gamma\rho}\ \bar{g}_{\alpha\beta}\ \bar{g}_{\mu\nu}\ \partial_\gamma \tilde{g}^{\alpha\beta} + \frac{1}{2}\ \tilde{g}^{\gamma\rho}\ \bar{g}_{\mu\alpha}\ \bar{g}_{\nu\beta}\ \partial_\gamma \tilde{g}^{\alpha\beta} + \frac{1}{4}\ \delta_\nu{}^\rho\ \bar{g}_{\alpha\beta}\ \partial_\mu \tilde{g}^{\alpha\beta} - \frac{1}{2}\ \bar{g}_{\nu\alpha}\ \partial_\mu \tilde{g}^{\alpha\rho} + \frac{1}{4}\ \delta_\mu{}^\rho\ \bar{g}_{\alpha\beta}\ \partial_\nu \tilde{g}^{\alpha\beta} - \frac{1}{2}\ \bar{g}_{\mu\alpha}\ \partial_\nu \tilde{g}^{\alpha\rho}$

Riemann tensor

Since now we have a formula for the Christoffel symbols, it is easy to calculate curvature expressions. Here we show this for the Riemann tensor. We know how it is expressed in terms of the

Christoffel symbols, and so we can use the formula above. Also we can calculate derivatives, since we know how they act on the gothic metric and its inverse.

In[33]:=
```
RiemannCD[-σ, -ρ, -ν, μ]
ChangeCurvature[%, CD, PD]
% /. chru;
% /. dgru;
Expand[%];
% //. ggru;
ToCanonical[%, UseMetricOnVBundle → None]
```

Out[33]= $R[\nabla]_{\sigma\rho\nu}{}^{\mu}$

Out[34]= $\Gamma[\nabla]^{\alpha}{}_{\sigma\nu}\,\Gamma[\nabla]^{\mu}{}_{\rho\alpha} - \Gamma[\nabla]^{\alpha}{}_{\rho\nu}\,\Gamma[\nabla]^{\mu}{}_{\sigma\alpha} + \partial_{\rho}\Gamma[\nabla]^{\mu}{}_{\sigma\nu} - \partial_{\sigma}\Gamma[\nabla]^{\mu}{}_{\rho\nu}$

Out[39]=

$$
\begin{aligned}
&-\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\nu\sigma}\partial_\alpha\partial_\rho\tilde{g}^{\beta\gamma}+\frac{1}{2}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\nu\beta}\underset{\sim}{\bar{g}}_{\sigma\gamma}\partial_\alpha\partial_\rho\tilde{g}^{\beta\gamma}+\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\nu\rho}\partial_\alpha\partial_\sigma\tilde{g}^{\beta\gamma}-\\
&\frac{1}{2}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\nu\beta}\underset{\sim}{\bar{g}}_{\rho\gamma}\partial_\alpha\partial_\sigma\tilde{g}^{\beta\gamma}+\frac{1}{4}\tilde{g}^{\alpha\beta}\underset{\sim}{\bar{g}}_{\nu\gamma}\underset{\sim}{\bar{g}}_{\rho\zeta}\underset{\sim}{\bar{g}}_{\sigma\theta}\partial_\alpha\tilde{g}^{\mu\theta}\partial_\beta\tilde{g}^{\gamma\zeta}-\\
&\frac{1}{4}\tilde{g}^{\alpha\beta}\underset{\sim}{\bar{g}}_{\nu\gamma}\underset{\sim}{\bar{g}}_{\rho\zeta}\underset{\sim}{\bar{g}}_{\sigma\theta}\partial_\alpha\tilde{g}^{\mu\zeta}\partial_\beta\tilde{g}^{\gamma\theta}+\frac{1}{8}\tilde{g}^{\alpha\beta}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\sigma}\underset{\sim}{\bar{g}}_{\rho\gamma}\partial_\alpha\tilde{g}^{\mu\gamma}\partial_\beta\tilde{g}^{\zeta\theta}-\\
&\frac{1}{8}\tilde{g}^{\alpha\beta}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\rho}\underset{\sim}{\bar{g}}_{\sigma\gamma}\partial_\alpha\tilde{g}^{\mu\gamma}\partial_\beta\tilde{g}^{\zeta\theta}+\frac{1}{16}\delta^\mu{}_\sigma\tilde{g}^{\alpha\beta}\underset{\sim}{\bar{g}}_{\gamma\zeta}\underset{\sim}{\bar{g}}_{\theta\vartheta}\underset{\sim}{\bar{g}}_{\nu\rho}\partial_\alpha\tilde{g}^{\gamma\zeta}\partial_\beta\tilde{g}^{\theta\vartheta}-\\
&\frac{1}{16}\delta^\mu{}_\rho\tilde{g}^{\alpha\beta}\underset{\sim}{\bar{g}}_{\gamma\zeta}\underset{\sim}{\bar{g}}_{\theta\vartheta}\underset{\sim}{\bar{g}}_{\nu\sigma}\partial_\alpha\tilde{g}^{\gamma\zeta}\partial_\beta\tilde{g}^{\theta\vartheta}-\frac{1}{8}\delta^\mu{}_\sigma\tilde{g}^{\alpha\beta}\underset{\sim}{\bar{g}}_{\theta\vartheta}\underset{\sim}{\bar{g}}_{\nu\gamma}\underset{\sim}{\bar{g}}_{\rho\zeta}\partial_\alpha\tilde{g}^{\gamma\zeta}\partial_\beta\tilde{g}^{\theta\vartheta}+\\
&\frac{1}{8}\delta^\mu{}_\rho\tilde{g}^{\alpha\beta}\underset{\sim}{\bar{g}}_{\theta\vartheta}\underset{\sim}{\bar{g}}_{\nu\gamma}\underset{\sim}{\bar{g}}_{\sigma\zeta}\partial_\alpha\tilde{g}^{\gamma\zeta}\partial_\beta\tilde{g}^{\theta\vartheta}-\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\nu\beta}\underset{\sim}{\bar{g}}_{\rho\gamma}\underset{\sim}{\bar{g}}_{\sigma\zeta}\partial_\alpha\tilde{g}^{\zeta\theta}\partial_\theta\tilde{g}^{\beta\gamma}+\\
&\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\nu\beta}\underset{\sim}{\bar{g}}_{\rho\gamma}\underset{\sim}{\bar{g}}_{\sigma\zeta}\partial_\alpha\tilde{g}^{\gamma\theta}\partial_\theta\tilde{g}^{\beta\zeta}-\frac{1}{8}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\gamma\zeta}\underset{\sim}{\bar{g}}_{\nu\sigma}\underset{\sim}{\bar{g}}_{\rho\beta}\partial_\alpha\tilde{g}^{\beta\theta}\partial_\theta\tilde{g}^{\gamma\zeta}+\\
&\frac{1}{8}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\gamma\zeta}\underset{\sim}{\bar{g}}_{\nu\rho}\underset{\sim}{\bar{g}}_{\sigma\beta}\partial_\alpha\tilde{g}^{\beta\theta}\partial_\theta\tilde{g}^{\gamma\zeta}-\frac{1}{4}\underset{\sim}{\bar{g}}_{\rho\alpha}\underset{\sim}{\bar{g}}_{\sigma\beta}\partial_\gamma\tilde{g}^{\mu\beta}\partial_\nu\tilde{g}^{\alpha\gamma}+\\
&\frac{1}{8}\delta^\mu{}_\sigma\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\rho\alpha}\partial_\zeta\tilde{g}^{\beta\gamma}\partial_\nu\tilde{g}^{\alpha\zeta}-\frac{1}{8}\delta^\mu{}_\rho\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\sigma\alpha}\partial_\zeta\tilde{g}^{\beta\gamma}\partial_\nu\tilde{g}^{\alpha\zeta}+\frac{1}{4}\underset{\sim}{\bar{g}}_{\rho\alpha}\underset{\sim}{\bar{g}}_{\sigma\beta}\partial_\gamma\tilde{g}^{\mu\alpha}\partial_\nu\tilde{g}^{\beta\gamma}+\\
&\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\rho\beta}\underset{\sim}{\bar{g}}_{\sigma\gamma}\partial_\alpha\tilde{g}^{\gamma\theta}\partial_\nu\tilde{g}^{\beta\zeta}-\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\rho\beta}\underset{\sim}{\bar{g}}_{\sigma\gamma}\partial_\alpha\tilde{g}^{\beta\theta}\partial_\nu\tilde{g}^{\gamma\zeta}-\\
&\frac{1}{4}\underset{\sim}{\bar{g}}_{\nu\alpha}\underset{\sim}{\bar{g}}_{\sigma\beta}\partial_\gamma\tilde{g}^{\mu\beta}\partial_\rho\tilde{g}^{\alpha\gamma}+\frac{1}{2}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\sigma\alpha}\partial_\nu\tilde{g}^{\mu\beta}\partial_\rho\tilde{g}^{\alpha\gamma}+\frac{1}{8}\delta^\mu{}_\sigma\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\nu\alpha}\partial_\zeta\tilde{g}^{\beta\gamma}\partial_\rho\tilde{g}^{\alpha\zeta}+\\
&\frac{1}{16}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\sigma}\partial_\alpha\tilde{g}^{\zeta\theta}\partial_\rho\tilde{g}^{\beta\gamma}+\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\sigma}\partial_\alpha\tilde{g}^{\gamma\theta}\partial_\rho\tilde{g}^{\beta\zeta}-\\
&\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\beta}\underset{\sim}{\bar{g}}_{\sigma\gamma}\partial_\alpha\tilde{g}^{\gamma\theta}\partial_\rho\tilde{g}^{\beta\zeta}-\frac{1}{4}\delta^\mu{}_\sigma\underset{\sim}{\bar{g}}_{\alpha\beta}\underset{\sim}{\bar{g}}_{\gamma\zeta}\partial_\nu\tilde{g}^{\alpha\gamma}\partial_\rho\tilde{g}^{\beta\zeta}-\\
&\frac{1}{2}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\beta}\underset{\sim}{\bar{g}}_{\sigma\gamma}\partial_\alpha\tilde{g}^{\beta\theta}\partial_\rho\tilde{g}^{\gamma\zeta}-\frac{1}{16}\delta^\mu{}_\sigma\underset{\sim}{\bar{g}}_{\alpha\beta}\underset{\sim}{\bar{g}}_{\gamma\zeta}\partial_\nu\tilde{g}^{\alpha\beta}\partial_\rho\tilde{g}^{\gamma\zeta}+\\
&\frac{1}{4}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\sigma\alpha}\partial_\nu\tilde{g}^{\alpha\gamma}\partial_\rho\tilde{g}^{\mu\beta}-\frac{1}{8}\underset{\sim}{\bar{g}}_{\alpha\beta}\underset{\sim}{\bar{g}}_{\nu\sigma}\partial_\gamma\tilde{g}^{\alpha\beta}\partial_\rho\tilde{g}^{\mu\gamma}+\frac{1}{4}\underset{\sim}{\bar{g}}_{\nu\alpha}\underset{\sim}{\bar{g}}_{\sigma\beta}\partial_\gamma\tilde{g}^{\alpha\beta}\partial_\rho\tilde{g}^{\mu\gamma}+\\
&\frac{1}{4}\delta^\mu{}_\sigma\underset{\sim}{\bar{g}}_{\alpha\beta}\partial_\rho\partial_\nu\tilde{g}^{\alpha\beta}-\frac{1}{2}\underset{\sim}{\bar{g}}_{\sigma\alpha}\partial_\rho\partial_\nu\tilde{g}^{\mu\alpha}+\frac{1}{4}\underset{\sim}{\bar{g}}_{\nu\alpha}\underset{\sim}{\bar{g}}_{\rho\beta}\partial_\gamma\tilde{g}^{\mu\beta}\partial_\sigma\tilde{g}^{\alpha\gamma}-\\
&\frac{1}{2}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\rho\alpha}\partial_\nu\tilde{g}^{\mu\beta}\partial_\sigma\tilde{g}^{\alpha\gamma}-\frac{1}{4}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\nu\alpha}\partial_\rho\tilde{g}^{\mu\beta}\partial_\sigma\tilde{g}^{\alpha\gamma}-\frac{1}{8}\delta^\mu{}_\rho\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\nu\alpha}\partial_\zeta\tilde{g}^{\beta\gamma}\partial_\sigma\tilde{g}^{\alpha\zeta}-\\
&\frac{1}{16}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\rho}\partial_\alpha\tilde{g}^{\zeta\theta}\partial_\sigma\tilde{g}^{\beta\gamma}-\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\rho}\partial_\alpha\tilde{g}^{\gamma\theta}\partial_\sigma\tilde{g}^{\beta\zeta}+\\
&\frac{1}{4}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\beta}\underset{\sim}{\bar{g}}_{\rho\gamma}\partial_\alpha\tilde{g}^{\gamma\theta}\partial_\sigma\tilde{g}^{\beta\zeta}+\frac{1}{4}\delta^\mu{}_\rho\underset{\sim}{\bar{g}}_{\alpha\beta}\underset{\sim}{\bar{g}}_{\gamma\zeta}\partial_\nu\tilde{g}^{\alpha\gamma}\partial_\sigma\tilde{g}^{\beta\zeta}+\\
&\frac{1}{2}\tilde{g}^{\mu\alpha}\underset{\sim}{\bar{g}}_{\zeta\theta}\underset{\sim}{\bar{g}}_{\nu\beta}\underset{\sim}{\bar{g}}_{\rho\gamma}\partial_\alpha\tilde{g}^{\beta\theta}\partial_\sigma\tilde{g}^{\gamma\zeta}+\frac{1}{16}\delta^\mu{}_\rho\underset{\sim}{\bar{g}}_{\alpha\beta}\underset{\sim}{\bar{g}}_{\gamma\zeta}\partial_\nu\tilde{g}^{\alpha\beta}\partial_\sigma\tilde{g}^{\gamma\zeta}-\\
&\frac{1}{4}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\rho\alpha}\partial_\nu\tilde{g}^{\alpha\gamma}\partial_\sigma\tilde{g}^{\mu\beta}+\frac{1}{4}\underset{\sim}{\bar{g}}_{\beta\gamma}\underset{\sim}{\bar{g}}_{\nu\alpha}\partial_\rho\tilde{g}^{\alpha\gamma}\partial_\sigma\tilde{g}^{\mu\beta}+\frac{1}{8}\underset{\sim}{\bar{g}}_{\alpha\beta}\underset{\sim}{\bar{g}}_{\nu\rho}\partial_\gamma\tilde{g}^{\alpha\beta}\partial_\sigma\tilde{g}^{\mu\gamma}-\\
&\frac{1}{4}\underset{\sim}{\bar{g}}_{\nu\alpha}\underset{\sim}{\bar{g}}_{\rho\beta}\partial_\gamma\tilde{g}^{\alpha\beta}\partial_\sigma\tilde{g}^{\mu\gamma}-\frac{1}{4}\delta^\mu{}_\rho\underset{\sim}{\bar{g}}_{\alpha\beta}\partial_\sigma\partial_\nu\tilde{g}^{\alpha\beta}+\frac{1}{2}\underset{\sim}{\bar{g}}_{\rho\alpha}\partial_\sigma\partial_\nu\tilde{g}^{\mu\alpha}
\end{aligned}
$$

Ricci tensor

With the knowledge from above, we can continue and calculate the Ricci tensor. Here we go one step further: we raise its indices with the gothic metric, and so we obtain a tensor density of weight +2. Note how density factors are fully absorbed in the gothic metric.

In[40]:=
```
DIMet[μ, α] × DIMet[ν, β] × RicciCD[-α, -β]
ChangeCurvature [%, CD, PD];
% /. chru;
% /. dgru;
Expand[%];
% //. ggru;
ToCanonical [%, UseMetricOnVBundle → None]
```

Out[40]= $\tilde{\mathfrak{g}}^{\mu\alpha}\,\tilde{\mathfrak{g}}^{\nu\beta}\,R[\nabla]_{\alpha\beta}$

Out[46]=
$$\frac{1}{2}\partial_\alpha\tilde{\mathfrak{g}}^{\mu\nu}\,\partial_\beta\tilde{\mathfrak{g}}^{\alpha\beta}+\frac{1}{4}\tilde{\mathfrak{g}}^{\alpha\beta}\,\tilde{\mathfrak{g}}^{\mu\nu}\,\bar{\mathfrak{g}}_{\gamma\zeta}\,\bar{\mathfrak{g}}_{\theta\vartheta}\,\partial_\alpha\tilde{\mathfrak{g}}^{\gamma\theta}\,\partial_\beta\tilde{\mathfrak{g}}^{\zeta\vartheta}-\frac{1}{4}\tilde{\mathfrak{g}}^{\mu\alpha}\,\tilde{\mathfrak{g}}^{\nu\beta}\,\bar{\mathfrak{g}}_{\gamma\zeta}\,\bar{\mathfrak{g}}_{\theta\vartheta}\,\partial_\alpha\tilde{\mathfrak{g}}^{\gamma\theta}\,\partial_\beta\tilde{\mathfrak{g}}^{\zeta\vartheta}+$$
$$\frac{1}{8}\tilde{\mathfrak{g}}^{\mu\alpha}\,\tilde{\mathfrak{g}}^{\nu\beta}\,\bar{\mathfrak{g}}_{\gamma\zeta}\,\bar{\mathfrak{g}}_{\theta\vartheta}\,\partial_\alpha\tilde{\mathfrak{g}}^{\gamma\zeta}\,\partial_\beta\tilde{\mathfrak{g}}^{\theta\vartheta}-\frac{1}{2}\partial_\alpha\tilde{\mathfrak{g}}^{\nu\beta}\,\partial_\beta\tilde{\mathfrak{g}}^{\mu\alpha}-\frac{1}{2}\tilde{\mathfrak{g}}^{\alpha\beta}\,\bar{\mathfrak{g}}_{\gamma\zeta}\,\partial_\alpha\tilde{\mathfrak{g}}^{\mu\gamma}\,\partial_\beta\tilde{\mathfrak{g}}^{\nu\zeta}-$$
$$\frac{1}{4}\tilde{\mathfrak{g}}^{\alpha\beta}\,\tilde{\mathfrak{g}}^{\mu\nu}\,\bar{\mathfrak{g}}_{\gamma\zeta}\,\partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\gamma\zeta}-\frac{1}{2}\tilde{\mathfrak{g}}^{\nu\alpha}\,\partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\mu\beta}+\frac{1}{2}\tilde{\mathfrak{g}}^{\alpha\beta}\,\partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\mu\nu}-\frac{1}{2}\tilde{\mathfrak{g}}^{\mu\alpha}\,\partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\nu\beta}-$$
$$\frac{1}{4}\tilde{\mathfrak{g}}^{\mu\nu}\,\bar{\mathfrak{g}}_{\alpha\beta}\,\partial_\gamma\tilde{\mathfrak{g}}^{\alpha\beta}\,\partial_\zeta\tilde{\mathfrak{g}}^{\gamma\zeta}+\frac{1}{2}\tilde{\mathfrak{g}}^{\nu\alpha}\,\bar{\mathfrak{g}}_{\beta\gamma}\,\partial_\alpha\tilde{\mathfrak{g}}^{\gamma\zeta}\,\partial_\zeta\tilde{\mathfrak{g}}^{\mu\beta}+\frac{1}{2}\tilde{\mathfrak{g}}^{\mu\alpha}\,\bar{\mathfrak{g}}_{\beta\gamma}\,\partial_\alpha\tilde{\mathfrak{g}}^{\gamma\zeta}\,\partial_\zeta\tilde{\mathfrak{g}}^{\nu\beta}$$

Einstein tensor

In the same spirit, we also raise the indices of the Einstein tensor with the gothic metric. Note here in particular the second term, where we have combined the density factors with the metric that is used to obtain the Ricci scalar. The result is not too lengthy.

In[47]:=
```
DIMet[μ, α] × DIMet[ν, β] × RicciCD[-α, -β] - RicciCD[-ρ, -σ] × DIMet[ρ, σ] × DIMet[μ, ν] / 2
ChangeCurvature [%, CD, PD];
% /. chru;
% /. dgru;
Expand[%];
% //. ggru;
eins = ToCanonical [%, UseMetricOnVBundle → None]
```

Out[47]=
$$\tilde{\mathfrak{g}}^{\mu\alpha}\ \tilde{\mathfrak{g}}^{\nu\beta}\ R[\nabla]_{\alpha\beta} - \frac{1}{2}\ \tilde{\mathfrak{g}}^{\mu\nu}\ \tilde{\mathfrak{g}}^{\rho\sigma}\ R[\nabla]_{\rho\sigma}$$

Out[53]=
$$\frac{1}{2}\partial_\alpha\tilde{\mathfrak{g}}^{\mu\nu}\ \partial_\beta\tilde{\mathfrak{g}}^{\alpha\beta} - \frac{1}{16}\ \tilde{\mathfrak{g}}^{\alpha\beta}\ \tilde{\mathfrak{g}}^{\mu\nu}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\gamma\zeta}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\rho\sigma}\ \partial_\alpha\tilde{\mathfrak{g}}^{\rho\sigma}\ \partial_\beta\tilde{\mathfrak{g}}^{\gamma\zeta} +$$
$$\frac{1}{8}\ \tilde{\mathfrak{g}}^{\mu\alpha}\ \tilde{\mathfrak{g}}^{\nu\beta}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\gamma\zeta}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\rho\sigma}\ \partial_\alpha\tilde{\mathfrak{g}}^{\rho\sigma}\ \partial_\beta\tilde{\mathfrak{g}}^{\gamma\zeta} - \frac{1}{2}\partial_\alpha\tilde{\mathfrak{g}}^{\nu\beta}\ \partial_\beta\tilde{\mathfrak{g}}^{\mu\alpha} - \frac{1}{2}\ \tilde{\mathfrak{g}}^{\alpha\beta}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\rho\sigma}\ \partial_\alpha\tilde{\mathfrak{g}}^{\mu\rho}\ \partial_\beta\tilde{\mathfrak{g}}^{\nu\sigma} +$$
$$\frac{1}{8}\ \tilde{\mathfrak{g}}^{\alpha\beta}\ \tilde{\mathfrak{g}}^{\mu\nu}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\gamma\zeta}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\rho\sigma}\ \partial_\alpha\tilde{\mathfrak{g}}^{\rho\gamma}\ \partial_\beta\tilde{\mathfrak{g}}^{\sigma\zeta} - \frac{1}{4}\ \tilde{\mathfrak{g}}^{\mu\alpha}\ \tilde{\mathfrak{g}}^{\nu\beta}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\gamma\zeta}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\rho\sigma}\ \partial_\alpha\tilde{\mathfrak{g}}^{\rho\gamma}\ \partial_\beta\tilde{\mathfrak{g}}^{\sigma\zeta} +$$
$$\frac{1}{2}\ \tilde{\mathfrak{g}}^{\mu\nu}\ \partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\alpha\beta} - \frac{1}{2}\ \tilde{\mathfrak{g}}^{\nu\alpha}\ \partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\mu\beta} + \frac{1}{2}\ \tilde{\mathfrak{g}}^{\alpha\beta}\ \partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\mu\nu} - \frac{1}{2}\ \tilde{\mathfrak{g}}^{\mu\alpha}\ \partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\nu\beta} -$$
$$\frac{1}{4}\ \tilde{\mathfrak{g}}^{\mu\nu}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\alpha\beta}\ \partial_\rho\tilde{\mathfrak{g}}^{\beta\sigma}\ \partial_\sigma\tilde{\mathfrak{g}}^{\alpha\rho} + \frac{1}{2}\ \tilde{\mathfrak{g}}^{\nu\alpha}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\beta\rho}\ \partial_\alpha\tilde{\mathfrak{g}}^{\rho\sigma}\ \partial_\sigma\tilde{\mathfrak{g}}^{\mu\beta} + \frac{1}{2}\ \tilde{\mathfrak{g}}^{\mu\alpha}\ \underset{\sim}{\bar{\mathfrak{g}}}_{\beta\rho}\ \partial_\alpha\tilde{\mathfrak{g}}^{\rho\sigma}\ \partial_\sigma\tilde{\mathfrak{g}}^{\nu\beta}$$

Harmonic gauge

Gauge condition

Recall that the harmonic gauge condition can be written as demanding that the contraction of the lower indices of the Christoffel symbols with the metric vanishes. Here we can equivalently use the gothic metric, which differs from the normal metric just by a density factor, which does not alter the equation. We see that the harmonic gauge condition just becomes a condition on the divergence.

In[54]:=
```
DIMet[μ, ν] × ChristoffelCD [ρ, -μ, -ν]
% /. chru;
Expand[%];
% //. ggru;
ToCanonical [%, UseMetricOnVBundle → None]
```

Out[54]=
$$\Gamma[\nabla]^{\rho}{}_{\mu\nu}\ \tilde{\mathfrak{g}}^{\mu\nu}$$

Out[58]=
$$-\partial_\mu\tilde{\mathfrak{g}}^{\rho\mu}$$

We implement this condition as a rule. We also add a few higher derivatives, so that the rule finds them more easily.

In[59]:=
```
hgru = Flatten[MakeRule[{#, 0}, MetricOn → None, ContractMetrics → False] & /@
    {PD[-α][DIMet[α, β]], PD[-α][PD[-γ][DIMet[α, β]]], PD[-α][PD[-γ][PD[-ζ][DIMet[α, β]]]]}]
```

Out[59]=
$$\left\{\text{HoldPattern}\left[\partial_{\underline{\alpha}}\tilde{\mathfrak{g}}^{\underline{\underline{\alpha}}\underline{\underline{\beta}}}\right] :\to \text{Module}[\{\}, 0],\ \text{HoldPattern}\left[\partial_{\underline{\alpha}}\tilde{\mathfrak{g}}^{\underline{\underline{\beta}}\underline{\underline{\alpha}}}\right] :\to \text{Module}[\{\}, 0],\right.$$
$$\text{HoldPattern}\left[\partial_{\underline{\alpha}}\partial_{\underline{\gamma}}\tilde{\mathfrak{g}}^{\underline{\underline{\alpha}}\underline{\underline{\beta}}}\right] :\to \text{Module}[\{\}, 0],\ \text{HoldPattern}\left[\partial_{\underline{\alpha}}\partial_{\underline{\gamma}}\tilde{\mathfrak{g}}^{\underline{\underline{\beta}}\underline{\underline{\alpha}}}\right] :\to \text{Module}[\{\}, 0],$$
$$\left.\text{HoldPattern}\left[\partial_{\underline{\alpha}}\partial_{\underline{\gamma}}\partial_{\underline{\zeta}}\tilde{\mathfrak{g}}^{\underline{\underline{\alpha}}\underline{\underline{\beta}}}\right] :\to \text{Module}[\{\}, 0],\ \text{HoldPattern}\left[\partial_{\underline{\alpha}}\partial_{\underline{\gamma}}\partial_{\underline{\zeta}}\tilde{\mathfrak{g}}^{\underline{\underline{\beta}}\underline{\underline{\alpha}}}\right] :\to \text{Module}[\{\}, 0]\right\}$$

A simple test shows that we are right.

In[60]:=
```
DIMet[μ, ν] × ChristoffelCD[ρ, -μ, -ν]
% /. chru;
Expand[%];
% //. ggru;
% /. hgru;
ToCanonical[%, UseMetricOnVBundle → None]
```

Out[60]=
$$\Gamma[\nabla]^{\rho}{}_{\mu\nu}\ \tilde{\mathfrak{g}}^{\mu\nu}$$

Out[65]=
$$0$$

Einstein tensor

Now we apply the harmonic gauge condition to the Einstein tensor. This indeed simplifies the result. Note that only one term with second order derivatives remains, which is just given by the (densitized) d'Alembert operator.

In[66]:=
```
einshg = eins /. hgru
```

Out[66]=
$$-\frac{1}{16}\tilde{\mathfrak{g}}^{\alpha\beta}\tilde{\mathfrak{g}}^{\mu\nu}\underset{\sim}{\mathfrak{g}}_{\gamma\zeta}\underset{\sim}{\mathfrak{g}}_{\rho\sigma}\partial_\alpha\tilde{\mathfrak{g}}^{\rho\sigma}\partial_\beta\tilde{\mathfrak{g}}^{\gamma\zeta}+\frac{1}{8}\tilde{\mathfrak{g}}^{\mu\alpha}\tilde{\mathfrak{g}}^{\nu\beta}\underset{\sim}{\mathfrak{g}}_{\gamma\zeta}\underset{\sim}{\mathfrak{g}}_{\rho\sigma}\partial_\alpha\tilde{\mathfrak{g}}^{\rho\sigma}\partial_\beta\tilde{\mathfrak{g}}^{\gamma\zeta}-$$
$$\frac{1}{2}\partial_\alpha\tilde{\mathfrak{g}}^{\nu\beta}\partial_\beta\tilde{\mathfrak{g}}^{\mu\alpha}-\frac{1}{2}\tilde{\mathfrak{g}}^{\alpha\beta}\underset{\sim}{\mathfrak{g}}_{\rho\sigma}\partial_\alpha\tilde{\mathfrak{g}}^{\mu\rho}\partial_\beta\tilde{\mathfrak{g}}^{\nu\sigma}+\frac{1}{8}\tilde{\mathfrak{g}}^{\alpha\beta}\tilde{\mathfrak{g}}^{\mu\nu}\underset{\sim}{\mathfrak{g}}_{\gamma\zeta}\underset{\sim}{\mathfrak{g}}_{\rho\sigma}\partial_\alpha\tilde{\mathfrak{g}}^{\rho\gamma}\partial_\beta\tilde{\mathfrak{g}}^{\sigma\zeta}-$$
$$\frac{1}{4}\tilde{\mathfrak{g}}^{\mu\alpha}\tilde{\mathfrak{g}}^{\nu\beta}\underset{\sim}{\mathfrak{g}}_{\gamma\zeta}\underset{\sim}{\mathfrak{g}}_{\rho\sigma}\partial_\alpha\tilde{\mathfrak{g}}^{\rho\gamma}\partial_\beta\tilde{\mathfrak{g}}^{\sigma\zeta}+\frac{1}{2}\tilde{\mathfrak{g}}^{\alpha\beta}\partial_\beta\partial_\alpha\tilde{\mathfrak{g}}^{\mu\nu}-$$
$$\frac{1}{4}\tilde{\mathfrak{g}}^{\mu\nu}\underset{\sim}{\mathfrak{g}}_{\alpha\beta}\partial_\rho\tilde{\mathfrak{g}}^{\beta\sigma}\partial_\sigma\tilde{\mathfrak{g}}^{\alpha\rho}+\frac{1}{2}\tilde{\mathfrak{g}}^{\nu\alpha}\underset{\sim}{\mathfrak{g}}_{\beta\rho}\partial_\alpha\tilde{\mathfrak{g}}^{\rho\sigma}\partial_\sigma\tilde{\mathfrak{g}}^{\mu\beta}+\frac{1}{2}\tilde{\mathfrak{g}}^{\mu\alpha}\underset{\sim}{\mathfrak{g}}_{\beta\rho}\partial_\alpha\tilde{\mathfrak{g}}^{\rho\sigma}\partial_\sigma\tilde{\mathfrak{g}}^{\nu\beta}$$

Einstein equations

Together with the energy-momentum tensor, we get the Einstein equations. Note that since we raised the indices with the gothic metric, and obtained a density of weight +2, we must respect this also for the energy-momentum tensor. Here we multiply with $-\det g$, and then move all factors to the geometry side.

In[67]:=
```
einsru = MakeRule[{EnMom[μ, ν], Evaluate[-einshg / 8 / Pi / NewtG / DetMet[]]},
    MetricOn → None, ContractMetrics → False];
```

This is what the equations look like. We check our rule, which makes them vanish.

In[68]:=
```
2 (einshg + 8 Pi NewtG EnMom[μ, ν] × DetMet[]);
feqhg = Expand[%]
% /. einsru;
Expand[%]
```

Out[69]=

$$16\,G\,\pi\,\tilde{\tilde{g}}\,T^{\mu\nu} - \frac{1}{8}\tilde{g}^{\alpha\beta}\,\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\beta\tilde{g}^{\gamma\zeta} + \frac{1}{4}\tilde{g}^{\mu\alpha}\,\tilde{g}^{\nu\beta}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\beta\tilde{g}^{\gamma\zeta} - \partial_\alpha\tilde{g}^{\nu\beta}\,\partial_\beta\tilde{g}^{\mu\alpha} - \tilde{g}^{\alpha\beta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\mu\rho}\,\partial_\beta\tilde{g}^{\nu\sigma} + \frac{1}{4}\tilde{g}^{\alpha\beta}\,\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\gamma}\,\partial_\beta\tilde{g}^{\sigma\zeta} - \frac{1}{2}\tilde{g}^{\mu\alpha}\,\tilde{g}^{\nu\beta}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\gamma}\,\partial_\beta\tilde{g}^{\sigma\zeta} + \tilde{g}^{\alpha\beta}\,\partial_\beta\partial_\alpha\tilde{g}^{\mu\nu} - \frac{1}{2}\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\alpha\beta}\,\partial_\rho\tilde{g}^{\beta\sigma}\,\partial_\sigma\tilde{g}^{\alpha\rho} + \tilde{g}^{\nu\alpha}\,\underset{\sim}{\bar{g}}_{\beta\rho}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\sigma\tilde{g}^{\mu\beta} + \tilde{g}^{\mu\alpha}\,\underset{\sim}{\bar{g}}_{\beta\rho}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\sigma\tilde{g}^{\nu\beta}$$

Out[71]=

$$-\frac{1}{8}\tilde{g}^{\alpha\beta}\,\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\beta\tilde{g}^{\gamma\zeta} + \frac{1}{4}\tilde{g}^{\mu\alpha}\,\tilde{g}^{\nu\beta}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\beta\tilde{g}^{\gamma\zeta} - \frac{1}{4}\tilde{g}^{\alpha\beta}\,\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\theta\vartheta}\,\partial_\alpha\tilde{g}^{\gamma\theta}\,\partial_\beta\tilde{g}^{\zeta\vartheta} + \frac{1}{2}\tilde{g}^{\mu\alpha}\,\tilde{g}^{\nu\beta}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\theta\vartheta}\,\partial_\alpha\tilde{g}^{\gamma\theta}\,\partial_\beta\tilde{g}^{\zeta\vartheta} + \frac{1}{8}\tilde{g}^{\alpha\beta}\,\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\theta\vartheta}\,\partial_\alpha\tilde{g}^{\gamma\zeta}\,\partial_\beta\tilde{g}^{\theta\vartheta} - \frac{1}{4}\tilde{g}^{\mu\alpha}\,\tilde{g}^{\nu\beta}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\theta\vartheta}\,\partial_\alpha\tilde{g}^{\gamma\zeta}\,\partial_\beta\tilde{g}^{\theta\vartheta} + \tilde{g}^{\alpha\beta}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\partial_\alpha\tilde{g}^{\mu\gamma}\,\partial_\beta\tilde{g}^{\nu\zeta} - \tilde{g}^{\alpha\beta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\mu\rho}\,\partial_\beta\tilde{g}^{\nu\sigma} + \frac{1}{4}\tilde{g}^{\alpha\beta}\,\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\gamma}\,\partial_\beta\tilde{g}^{\sigma\zeta} - \frac{1}{2}\tilde{g}^{\mu\alpha}\,\tilde{g}^{\nu\beta}\,\underset{\sim}{\bar{g}}_{\gamma\zeta}\,\underset{\sim}{\bar{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\gamma}\,\partial_\beta\tilde{g}^{\sigma\zeta} + \frac{1}{2}\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\alpha\beta}\,\partial_\gamma\tilde{g}^{\beta\zeta}\,\partial_\zeta\tilde{g}^{\alpha\gamma} - \tilde{g}^{\nu\alpha}\,\underset{\sim}{\bar{g}}_{\beta\gamma}\,\partial_\alpha\tilde{g}^{\gamma\zeta}\,\partial_\zeta\tilde{g}^{\mu\beta} - \tilde{g}^{\mu\alpha}\,\underset{\sim}{\bar{g}}_{\beta\gamma}\,\partial_\alpha\tilde{g}^{\gamma\zeta}\,\partial_\zeta\tilde{g}^{\nu\beta} - \frac{1}{2}\tilde{g}^{\mu\nu}\,\underset{\sim}{\bar{g}}_{\alpha\beta}\,\partial_\rho\tilde{g}^{\beta\sigma}\,\partial_\sigma\tilde{g}^{\alpha\rho} + \tilde{g}^{\nu\alpha}\,\underset{\sim}{\bar{g}}_{\beta\rho}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\sigma\tilde{g}^{\mu\beta} + \tilde{g}^{\mu\alpha}\,\underset{\sim}{\bar{g}}_{\beta\rho}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\sigma\tilde{g}^{\nu\beta}$$

Energy-momentum conservation

Covariant conservation

Next, we check how energy-momentum conservation can be found in the field equations. Here we define a rule which applies the covariant conservation law by replacing the non-covariant divergence of the energy-momentum tensor by Christoffel symbols.

In[72]:=
```
consru = MakeRule[{PD[-μ][EnMom[μ, ν]],
   -ChristoffelCD[μ, -μ, -ρ] × EnMom[ρ, ν] - ChristoffelCD[ν, -μ, -ρ] × EnMom[μ, ρ]},
  MetricOn → None, ContractMetrics → False]
```

Out[72]=

$$\{\mathrm{HoldPattern}[\partial_{\underline{\mu}} T^{\underline{\mu}\,\underline{\nu}}] :\rightarrow \mathrm{Module}[\{\alpha, \beta\}, -\Gamma[\nabla]^{\alpha}{}_{\alpha\beta}\,T^{\beta\nu} - \Gamma[\nabla]^{\nu}{}_{\alpha\beta}\,T^{\alpha\beta}],$$
$$\mathrm{HoldPattern}[\partial_{\underline{\mu}} T^{\underline{\nu}\,\underline{\mu}}] :\rightarrow \mathrm{Module}[\{\alpha, \beta\}, -\Gamma[\nabla]^{\alpha}{}_{\alpha\beta}\,T^{\beta\nu} - \Gamma[\nabla]^{\nu}{}_{\alpha\beta}\,T^{\alpha\beta}]\}$$

Metric substitutions

In the next step, we will encounter derivatives of the metric. Also those turn into Christoffel symbols.

In[73]:=
```
dmru = MakeRule[{PD[-ρ][Met[-μ, -ν]],
    ChristoffelCD[σ, -ρ, -μ] × Met[-σ, -ν] + ChristoffelCD[σ, -ρ, -ν] × Met[-μ, -σ]},
  MetricOn → None, ContractMetrics → False]
```

Out[73]= $\{\text{HoldPattern}[\partial_{\rho} g_{\mu\nu}] :\to \text{Module}[\{\alpha\}, \Gamma[\nabla]^{\alpha}{}_{\rho\mu}\ g_{\alpha\nu} + \Gamma[\nabla]^{\alpha}{}_{\rho\nu}\ g_{\mu\alpha}]\}$

Finally, we will replace the appearing metric by the gothic one.

In[74]:=
```
mgru = Flatten[MakeRule[#, MetricOn → None, ContractMetrics → False] & /@
    {{Met[μ, ν], DIMet[μ, ν] / Sqrt[-DetMet[]]}, {Met[-μ, -ν], DMet[-μ, -ν] Sqrt[-DetMet[]]}}]
```

Out[74]= $\{\text{HoldPattern}[g^{\mu\nu}] :\to \text{Module}[\{\}, \frac{\tilde{g}^{\mu\nu}}{\sqrt{-\tilde{\tilde{g}}}}], \text{HoldPattern}[g_{\mu\nu}] :\to \text{Module}[\{\}, \sqrt{-\tilde{\tilde{g}}}\ \tilde{g}_{\mu\nu}]\}$

Divergence equation

We now calculate the ordinary, non-covariant divergence of the field equations. Then we substitute the covariant energy-momentum conservation, to cancel the derivative acting on the energy-momentum tensor. We replace the remaining energy-momentum tensor by imposing the field equations. Also Christoffel symbols and the ordinary metric are replaced by the gothic metric. Finally, we impose the harmonic gauge. The result vanishes.

In[75]:=
```
PD[-μ][feqhg];
% /. consru /. dmru /. chru /. einsru /. mgru /. dgru;
Expand[%];
% //. ggru;
% /. hgru;
ToCanonical[%, UseMetricOnVBundle → None]
```

Out[80]= 0

Perturbative expansion

Set background to Minkowski metric

The full power of the harmonic expansion becomes apparent when we consider a perturbation $\mathfrak{g}^{\mu\nu} = \eta^{\mu\nu} + h^{\mu\nu}$ around the Minkowski metric. This implements the background.

In[81]:=
```
bgru = {DMet → Mink, DIMet → InvMink};
```

Perturbation of the inverse

We also use the formula for the n'th order perturbation of an inverse metric. Here it works just the same way for a density.

In[82]:=
```
ihh[LI[k_Integer /; k > 0], -μ_, -ν_] := Module[{ρ, σ, j},
  -DMet[-ν, -σ] × Sum[Binomial[k, j] ihh[LI[k - j], -μ, -ρ] × hh[LI[j], ρ, σ], {j, k}]]
```

These are the lowest orders.

In[83]:=
```
TableForm[ScreenDollarIndices[Expand[Table[ihh[LI[k], -μ, -ν], {k, 3}]]]]
```

Out[83]//TableForm=

$$-\bar{\tilde{g}}_{\mu\alpha}\,\bar{\tilde{g}}_{\nu\beta}\,h^{1\alpha\beta}$$

$$2\,\bar{\tilde{g}}_{\alpha\zeta}\,\bar{\tilde{g}}_{\mu\beta}\,\bar{\tilde{g}}_{\nu\gamma}\,h^{1\alpha\gamma}\,h^{1\beta\zeta}-\bar{\tilde{g}}_{\mu\alpha}\,\bar{\tilde{g}}_{\nu\beta}\,h^{2\alpha\beta}$$

$$-6\,\bar{\tilde{g}}_{\alpha\theta}\,\bar{\tilde{g}}_{\beta\vartheta}\,\bar{\tilde{g}}_{\mu\gamma}\,\bar{\tilde{g}}_{\nu\zeta}\,h^{1\alpha\zeta}\,h^{1\beta\theta}\,h^{1\gamma\vartheta}+3\,\bar{\tilde{g}}_{\alpha\zeta}\,\bar{\tilde{g}}_{\mu\beta}\,\bar{\tilde{g}}_{\nu\gamma}\,h^{1\beta\zeta}\,h^{2\alpha\gamma}+3\,\bar{\tilde{g}}_{\alpha\zeta}\,\bar{\tilde{g}}_{\mu\beta}\,\bar{\tilde{g}}_{\nu\gamma}\,h^{1\alpha\gamma}$$

Vacuum equations

For simplicity, we consider only the vacuum equations below.

In[84]:=
```
vfeqhg = feqhg /. {EnMom → Zero}
```

Out[84]=

$$-\frac{1}{8}\,\tilde{g}^{\alpha\beta}\,\tilde{g}^{\mu\nu}\,\bar{\tilde{g}}_{\gamma\zeta}\,\bar{\tilde{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\beta\tilde{g}^{\gamma\zeta}+\frac{1}{4}\,\tilde{g}^{\mu\alpha}\,\tilde{g}^{\nu\beta}\,\bar{\tilde{g}}_{\gamma\zeta}\,\bar{\tilde{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\beta\tilde{g}^{\gamma\zeta}-\partial_\alpha\tilde{g}^{\nu\beta}\,\partial_\beta\tilde{g}^{\mu\alpha}-\tilde{g}^{\alpha\beta}\,\bar{\tilde{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\mu\rho}\,\partial_\beta\tilde{g}^{\nu\sigma}+\frac{1}{4}\,\tilde{g}^{\alpha\beta}\,\tilde{g}^{\mu\nu}\,\bar{\tilde{g}}_{\gamma\zeta}\,\bar{\tilde{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\gamma}\,\partial_\beta\tilde{g}^{\sigma\zeta}-\frac{1}{2}\,\tilde{g}^{\mu\alpha}\,\tilde{g}^{\nu\beta}\,\bar{\tilde{g}}_{\gamma\zeta}\,\bar{\tilde{g}}_{\rho\sigma}\,\partial_\alpha\tilde{g}^{\rho\gamma}\,\partial_\beta\tilde{g}^{\sigma\zeta}+\tilde{g}^{\alpha\beta}\,\partial_\beta\partial_\alpha\tilde{g}^{\mu\nu}-\frac{1}{2}\,\tilde{g}^{\mu\nu}\,\bar{\tilde{g}}_{\alpha\beta}\,\partial_\rho\tilde{g}^{\beta\sigma}\,\partial_\sigma\tilde{g}^{\alpha\rho}+\tilde{g}^{\nu\alpha}\,\bar{\tilde{g}}_{\beta\rho}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\sigma\tilde{g}^{\mu\beta}+\tilde{g}^{\mu\alpha}\,\bar{\tilde{g}}_{\beta\rho}\,\partial_\alpha\tilde{g}^{\rho\sigma}\,\partial_\sigma\tilde{g}^{\nu\beta}$$

Perturbation around Minkowski metric

Now we are ready to come to the final result. We find that at each order in the expansion, we find a formula for the d’Alembertian of the corresponding perturbation, with products of lower order terms as the source. This is a general formula.

In[85]:=
```
ddη[k_] := (# == ToCanonical[Expand[# - Perturbation[vfeqhg, k] /. bgru],
      UseMetricOnVBundle → None]) &[InvMink[α, β] × PD[-β][PD[-α][hh[LI[k], μ, ν]]]]
```

At the first order, we just get the homogeneous wave equation. This we have seen already when we studied the linearized field equations alone.

In[86]:=
```
ddη[1]
```

Out[86]=

$$i\eta^{\alpha\beta}\,\partial_\beta\partial_\alpha h^{1\mu\nu}==0$$

At the second order, we find a source term, which is quadratic in the first order perturbation. We can thus solve it recursively.

In[87]:= **dd$\boldsymbol{\eta}$[2]**

Out[87]=
$$
\begin{aligned}
&\mathrm{i}\eta^{\alpha\beta}\ \partial_\beta\partial_\alpha \mathrm{h}^{2\mu\nu} == \\
&\frac{1}{4}\ \mathrm{i}\eta^{\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \eta_{\gamma\zeta}\ \eta_{\rho\sigma}\ \partial_\alpha \mathrm{h}^{1\rho\sigma}\ \partial_\beta \mathrm{h}^{1\gamma\zeta} - \frac{1}{2}\ \mathrm{i}\eta^{\mu\alpha}\ \mathrm{i}\eta^{\nu\beta}\ \eta_{\gamma\zeta}\ \eta_{\rho\sigma}\ \partial_\alpha \mathrm{h}^{1\rho\sigma}\ \partial_\beta \mathrm{h}^{1\gamma\zeta} + \\
&2\,\partial_\alpha \mathrm{h}^{1\nu\beta}\ \partial_\beta \mathrm{h}^{1\mu\alpha} + 2\ \mathrm{i}\eta^{\alpha\beta}\ \eta_{\rho\sigma}\ \partial_\alpha \mathrm{h}^{1\mu\rho}\ \partial_\beta \mathrm{h}^{1\nu\sigma} - \frac{1}{2}\ \mathrm{i}\eta^{\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \eta_{\gamma\zeta}\ \eta_{\rho\sigma}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\beta \mathrm{h}^{1\sigma\zeta} + \\
&\mathrm{i}\eta^{\mu\alpha}\ \mathrm{i}\eta^{\nu\beta}\ \eta_{\gamma\zeta}\ \eta_{\rho\sigma}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\beta \mathrm{h}^{1\sigma\zeta} - 2\ \mathrm{h}^{1\alpha\beta}\ \partial_\beta\partial_\alpha \mathrm{h}^{1\mu\nu} + \\
&\mathrm{i}\eta^{\mu\nu}\ \eta_{\alpha\beta}\ \partial_\rho \mathrm{h}^{1\beta\sigma}\ \partial_\sigma \mathrm{h}^{1\alpha\rho} - 2\ \mathrm{i}\eta^{\nu\alpha}\ \eta_{\beta\rho}\ \partial_\alpha \mathrm{h}^{1\rho\sigma}\ \partial_\sigma \mathrm{h}^{1\mu\beta} - 2\ \mathrm{i}\eta^{\mu\alpha}\ \eta_{\beta\rho}\ \partial_\alpha \mathrm{h}^{1\rho\sigma}\ \partial_\sigma \mathrm{h}^{1\nu\beta}
\end{aligned}
$$

At the third order, the terms become already more involved - but the scheme continues.

In[88]:= **dd$\boldsymbol{\eta}$[3]**

Out[88]=
$$
\begin{aligned}
&\mathrm{i}\eta^{\alpha\beta}\ \partial_\beta\partial_\alpha \mathrm{h}^{3\mu\nu} == \\
&-\frac{3}{2}\ \mathrm{h}^{1\mu\nu}\ \mathrm{i}\eta^{\alpha\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\zeta}\ \partial_\beta \mathrm{h}^{1\gamma\theta} + 3\ \mathrm{h}^{1\nu\alpha}\ \mathrm{i}\eta^{\mu\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\zeta}\ \partial_\beta \mathrm{h}^{1\gamma\theta} - \\
&\frac{3}{2}\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\zeta}\ \partial_\beta \mathrm{h}^{1\gamma\theta} + 3\ \mathrm{h}^{1\mu\alpha}\ \mathrm{i}\eta^{\nu\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\zeta}\ \partial_\beta \mathrm{h}^{1\gamma\theta} + \\
&\frac{3}{4}\ \mathrm{h}^{1\mu\nu}\ \mathrm{i}\eta^{\alpha\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\beta \mathrm{h}^{1\zeta\theta} - \frac{3}{2}\ \mathrm{h}^{1\nu\alpha}\ \mathrm{i}\eta^{\mu\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\beta \mathrm{h}^{1\zeta\theta} + \\
&\frac{3}{4}\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\beta \mathrm{h}^{1\zeta\theta} - \frac{3}{2}\ \mathrm{h}^{1\mu\alpha}\ \mathrm{i}\eta^{\nu\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\beta \mathrm{h}^{1\zeta\theta} + \\
&3\,\partial_\alpha \mathrm{h}^{2\nu\beta}\ \partial_\beta \mathrm{h}^{1\mu\alpha} + 6\ \mathrm{h}^{1\alpha\beta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\mu\rho}\ \partial_\beta \mathrm{h}^{1\nu\gamma} + 3\ \mathrm{i}\eta^{\alpha\beta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{2\mu\rho}\ \partial_\beta \mathrm{h}^{1\nu\gamma} - \\
&\frac{3}{4}\ \mathrm{i}\eta^{\mu\alpha}\ \mathrm{i}\eta^{\nu\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{2\zeta\theta}\ \partial_\beta \mathrm{h}^{1\rho\gamma} + \frac{3}{2}\ \mathrm{i}\eta^{\mu\alpha}\ \mathrm{i}\eta^{\nu\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{2\gamma\theta}\ \partial_\beta \mathrm{h}^{1\rho\zeta} - \\
&\frac{3}{2}\ \mathrm{i}\eta^{\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\zeta}\ \partial_\beta \mathrm{h}^{2\gamma\theta} + \frac{3}{2}\ \mathrm{i}\eta^{\mu\alpha}\ \mathrm{i}\eta^{\nu\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\zeta}\ \partial_\beta \mathrm{h}^{2\gamma\theta} + \\
&\frac{3}{4}\ \mathrm{i}\eta^{\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\beta \mathrm{h}^{2\zeta\theta} - \frac{3}{4}\ \mathrm{i}\eta^{\mu\alpha}\ \mathrm{i}\eta^{\nu\beta}\ \eta_{\zeta\theta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\beta \mathrm{h}^{2\zeta\theta} + \\
&3\,\partial_\alpha \mathrm{h}^{1\nu\beta}\ \partial_\beta \mathrm{h}^{2\mu\alpha} + 3\ \mathrm{i}\eta^{\alpha\beta}\ \eta_{\rho\gamma}\ \partial_\alpha \mathrm{h}^{1\mu\rho}\ \partial_\beta \mathrm{h}^{2\nu\gamma} - 3\ \mathrm{h}^{2\alpha\beta}\ \partial_\beta\partial_\alpha \mathrm{h}^{1\mu\nu} - \\
&3\ \mathrm{h}^{1\alpha\beta}\ \partial_\beta\partial_\alpha \mathrm{h}^{2\mu\nu} - 6\ \mathrm{h}^{1\nu\alpha}\ \eta_{\beta\rho}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\gamma \mathrm{h}^{1\mu\beta} - 3\ \mathrm{i}\eta^{\nu\alpha}\ \eta_{\beta\rho}\ \partial_\alpha \mathrm{h}^{2\rho\gamma}\ \partial_\gamma \mathrm{h}^{1\mu\beta} - \\
&6\ \mathrm{h}^{1\mu\alpha}\ \eta_{\beta\rho}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\gamma \mathrm{h}^{1\nu\beta} - 3\ \mathrm{i}\eta^{\mu\alpha}\ \eta_{\beta\rho}\ \partial_\alpha \mathrm{h}^{2\rho\gamma}\ \partial_\gamma \mathrm{h}^{1\nu\beta} - 3\ \mathrm{i}\eta^{\nu\alpha}\ \eta_{\beta\rho}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\gamma \mathrm{h}^{2\mu\beta} - \\
&3\ \mathrm{i}\eta^{\mu\alpha}\ \eta_{\beta\rho}\ \partial_\alpha \mathrm{h}^{1\rho\gamma}\ \partial_\gamma \mathrm{h}^{2\nu\beta} - 3\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \eta_{\alpha\rho}\ \eta_{\beta\gamma}\ \partial_\zeta \mathrm{h}^{1\gamma\theta}\ \partial_\theta \mathrm{h}^{1\rho\zeta} + \\
&3\ \mathrm{h}^{1\mu\nu}\ \eta_{\alpha\beta}\ \partial_\gamma \mathrm{h}^{1\alpha\rho}\ \partial_\rho \mathrm{h}^{1\beta\gamma} + \frac{3}{2}\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\rho}\ \mathrm{i}\eta^{\nu\gamma}\ \eta_{\alpha\zeta}\ \eta_{\beta\theta}\ \eta_{\vartheta\kappa}\ \partial_\gamma \mathrm{h}^{1\vartheta\kappa}\ \partial_\rho \mathrm{h}^{1\zeta\theta} - \\
&\frac{3}{2}\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \mathrm{i}\eta^{\rho\gamma}\ \eta_{\alpha\zeta}\ \eta_{\beta\theta}\ \eta_{\vartheta\kappa}\ \partial_\gamma \mathrm{h}^{1\vartheta\kappa}\ \partial_\rho \mathrm{h}^{1\zeta\theta} + \\
&6\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\nu\rho}\ \eta_{\alpha\gamma}\ \eta_{\beta\zeta}\ \partial_\theta \mathrm{h}^{1\mu\gamma}\ \partial_\rho \mathrm{h}^{1\zeta\theta} + 6\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\rho}\ \eta_{\alpha\gamma}\ \eta_{\beta\zeta}\ \partial_\theta \mathrm{h}^{1\nu\gamma}\ \partial_\rho \mathrm{h}^{1\zeta\theta} - \\
&6\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\rho}\ \mathrm{i}\eta^{\nu\gamma}\ \eta_{\alpha\zeta}\ \eta_{\beta\theta}\ \eta_{\vartheta\kappa}\ \partial_\gamma \mathrm{h}^{1\theta\kappa}\ \partial_\rho \mathrm{h}^{1\zeta\vartheta} + \\
&3\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\nu}\ \mathrm{i}\eta^{\rho\gamma}\ \eta_{\alpha\zeta}\ \eta_{\beta\theta}\ \eta_{\vartheta\kappa}\ \partial_\gamma \mathrm{h}^{1\theta\kappa}\ \partial_\rho \mathrm{h}^{1\zeta\vartheta} + \\
&\frac{3}{2}\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\mu\rho}\ \mathrm{i}\eta^{\nu\gamma}\ \eta_{\alpha\zeta}\ \eta_{\beta\theta}\ \eta_{\vartheta\kappa}\ \partial_\gamma \mathrm{h}^{1\zeta\theta}\ \partial_\rho \mathrm{h}^{1\vartheta\kappa} - \\
&6\ \mathrm{h}^{1\alpha\beta}\ \mathrm{i}\eta^{\rho\gamma}\ \eta_{\alpha\zeta}\ \eta_{\beta\theta}\ \partial_\gamma \mathrm{h}^{1\nu\theta}\ \partial_\rho \mathrm{h}^{1\mu\zeta} + 3\ \mathrm{i}\eta^{\mu\nu}\ \eta_{\alpha\beta}\ \partial_\gamma \mathrm{h}^{1\alpha\rho}\ \partial_\rho \mathrm{h}^{2\beta\gamma}
\end{aligned}
$$